

PROGRAMACIÓN DECLARATIVA

PROGRAMACIÓN LÓGICA

Tema PL2: El lenguaje Prolog, aspectos básicos

3. Semántica

Grado en Ingeniería Informática

URJC

Ana Pradera

Contenido

1 INTRODUCCIÓN

2 UNIFICACIÓN

- Sustituciones
- Unificadores
- El Algoritmo de Unificación
- Ejercicios de Unificación
- Unificación en PROLOG

3 LA REGLA DE RESOLUCIÓN

- Definición
- Ejemplos y ejercicios

4 EL ÁRBOL DE RESOLUCIÓN

- Construcción de Árboles de Resolución
- Mecanismo de cómputo de PROLOG

INTRODUCCIÓN

- En lo que sigue se describe la *semántica operacional* de PROLOG, es decir, su mecanismo de cómputo.
- PROLOG implementa un refinamiento del sistema de Demostración Automática conocido como **Sistema de Resolución SLD**.
- Para comprender su funcionamiento es necesario conocer:
 - El **problema de la Unificación y el algoritmo para resolverlo**.
 - La **Regla de Resolución**, que utiliza el Algoritmo de Unificación.
 - Los **Árboles de Resolución**, que utilizan la Regla de Resolución.

UNIFICACIÓN

Expresión = término (constante, variable o compuesto) o predicado

El problema de la Unificación

Dadas dos expresiones, el problema de la Unificación consiste en determinar si dichas expresiones pueden resultar sintácticamente idénticas mediante la realización de sustituciones adecuadas en sus variables, en cuyo caso se dice que las expresiones son unificables.

Ejemplos

- 1 $p(f(X), a)$ y $p(f(g(b)), Y)$ son unificables: basta sustituir X por $g(b)$ e Y por a , obteniéndose la expresión $p(f(g(b)), a)$.
- 2 $p(f(X), a)$ y $p(f(g(b)), c)$ NO son unificables: nunca podrán resultar iguales, al ser a y c dos constantes distintas.

Sustituciones

- Una **sustitución** es una función que permite asociar términos a variables.
- Las sustituciones se pueden **aplicar** a expresiones (términos o predicados), dando lugar a nuevas expresiones denominadas *instancias* de las primeras.
- Las sustituciones se pueden **componer** entre sí dando lugar a nuevas sustituciones.

Definición (Sustitución)

Una **sustitución** es un conjunto finito de asociaciones entre variables y términos, denotado $\{X_1/t_1, X_2/t_2, \dots, X_n/t_n\}$, donde:

- X_1, \dots, X_n son variables diferentes entre sí.
- t_1, \dots, t_n son términos y se verifica $t_i \neq X_i$ para todo $i \in \{1, \dots, n\}$.

Notación

- Cuando $n = 0$ se obtiene la *sustitución vacía*, denotada ϵ o $\{\}$.
- El resto de sustituciones se denotan σ, τ, θ , etc.

Ejemplos

- $\sigma = \{X/a, Y/f(a, b), Z/g(Y)\}$ y $\tau = \{X/f(a, Z), Y/g(f(a, Y))\}$ son sustituciones.
- $\{X/a, Y/f(a, b), X/g(Y)\}$ y $\{X/f(a, Z), Y/g(f(a, Y)), Z/Z\}$ no son sustituciones.

Definición (Aplicación de una sustitución a una expresión)

Sea φ una expresión y sea $\sigma = \{X_1/t_1, X_2/t_2, \dots, X_n/t_n\}$ una sustitución. La **aplicación de σ a φ** se define como la expresión resultante de sustituir en φ , de forma simultánea, cada aparición de la variable X_i , $1 \leq i \leq n$, por su término asociado en la sustitución σ , t_i . La expresión obtenida, denotada $\varphi\sigma$, se denomina una *instancia* de φ .

Ejemplos

- Si $\sigma = \{X/f(X, U), Y/X, Z/h(b, Y), U/a\}$ y $\varphi = g(X, Y, Z)$, entonces $\varphi\sigma = g(f(X, U), X, h(b, Y))$. ($\varphi\sigma \neq g(f(X, a), X, h(b, Y))$ y $\varphi\sigma \neq g(f(X, U), f(X, U), h(b, Y))$ puesto que *no* se encadenan las sustituciones, se hacen de forma simultánea).
- Si $\sigma = \{X/Y, Y/f(a)\}$ y $\varphi = p(X, f(Y), Z)$, entonces $\varphi\sigma = p(Y, f(f(a)), Z)$. ($\varphi\sigma \neq p(f(a), f(f(a)), Z)$ puesto que *no* se encadenan las sustituciones, se hacen de forma simultánea).

Definición (Composición de dos sustituciones)

Sean $\sigma = \{X_1/t_1, \dots, X_n/t_n\}$ y $\tau = \{Y_1/s_1, \dots, Y_m/s_m\}$ dos sustituciones. La **composición de σ con τ** , denotada $\sigma\tau$, es la sustitución compuesta por:

- 1 todos los pares $X_i/(t_i\tau)$, $1 \leq i \leq n$, salvo aquellos tales que $t_i\tau = X_j$ ($t_i\tau$ es la aplicación de τ a t_i).
- 2 todos los pares Y_j/s_j , $1 \leq j \leq m$, salvo aquellos tales que $Y_j \in \{X_1, \dots, X_n\}$.

Ejemplo

Sea $\sigma = \{X/g(U), Y/f(Z), Z/Y\}$ y $\tau = \{U/a, X/b, Y/Z, Z/g(X)\}$.

Entonces:

$$\begin{aligned}\sigma\tau &= \{X/g(a), Y/f(g(X)), \mathbf{Z/Z}, U/a, \mathbf{X/b}, \mathbf{Y/Z}, \mathbf{Z/g(X)}\} \\ &= \{X/g(a), Y/f(g(X)), U/a\}\end{aligned}$$

Ejercicios (Aplicación y composición de sustituciones)

1 Dadas las sustituciones $\sigma_1 = \{U/a, X/b, Y/Z, Z/g(X)\}$ y $\sigma_2 = \{X/g(U), Y/f(Z), Z/Y, V/U\}$:

- Calcule la composición $\sigma_1\sigma_2$.
- Aplique la sustitución resultante a la expresión $\varphi = p(X, f(Z), Y, V)$, es decir, calcule $\varphi\sigma_1\sigma_2$.

2 Dadas las sustituciones $\sigma_1 = \{X/Z, Z/g(Y, Y), V/Y, T/X\}$ y $\sigma_2 = \{Y/V, Z/g(V, V), T/a\}$:

- Calcule $\sigma_1\sigma_2$.
- Calcule $\sigma_2\sigma_1$.
- ¿Existe alguna sustitución θ tal que $\sigma_1\theta = \sigma_2$?
- ¿Existe alguna sustitución θ tal que $\sigma_2\theta = \sigma_1$?

Soluciones propuestas:

1.

- En el paso 1 de la definición de composición se obtienen los pares $\{U/(a\sigma_2), X/(b\sigma_2), Y/(Z\sigma_2), Z/(g(X)\sigma_2)\} = \{U/a, X/b, Y/Y, Z/g(g(U))\}$, de los que hay que eliminar Y/Y . En el paso 2, de los pares de σ_2 solo se añade V/U , puesto que las demás variables que aparecen a la izquierda en sus pares, X, Y, Z , son variables de σ_1 . Se obtiene por lo tanto $\sigma_1\sigma_2 = \{U/a, X/b, Z/g(g(U)), V/U\}$.
- Aplicando esta última sustitución a la expresión φ se obtiene $\varphi\sigma_1\sigma_2 = p(b, f(g(g(U))), Y, U)$.

2.

- $\sigma_1\sigma_2 = \{X/g(V, V), Z/g(V, V), T/X, Y/V\}$.
- $\sigma_2\sigma_1 = \{Z/g(Y, Y), T/a, X/Z, V/Y\}$.
- Sí: $\theta = \{Z/X, Y/V, X/a\}$.
 - El par Z/X es necesario para que el par X/Z de σ_1 desaparezca de $\sigma_1\theta$.
 - El par Y/V es necesario para transformar el par $Z/g(Y, Y)$ de σ_1 en el par $Z/g(V, V)$ de σ_2 .
 - El par X/a es necesario para transformar el par T/X de σ_1 en el par T/a de σ_2 .
- No, puesto que en el paso 1 de la composición de σ_2 con θ se deberá incluir siempre, sea quien sea θ , el par T/a (puesto que a es una constante y por lo tanto $a\theta = a$ para cualquier θ), y resulta que σ_1 no incluye dicho par.

Unificadores

Definición (Unificador)

Sean φ_1 y φ_2 dos expresiones. Si existe una sustitución σ tal que $\varphi_1\sigma = \varphi_2\sigma$, se dice que φ_1 y φ_2 son **unificables** y que la sustitución σ es un **unificador** de φ_1, φ_2 .

Expresiones	¿Unifican?
a, b	No
$a, f(X)$	No
X, b	Sí
$X, g(a, Y)$	Sí

Expresiones	¿Unifican?
$X, f(X, a)$	No
$p(a), p(f(Y))$	No
$g(a, b), g(U, U)$	No
$p(X, f(Y), Z), p(g(a), f(b), U)$	Sí

Observación

Cuando dos expresiones resultan ser unificables, es posible que la unificación se pueda conseguir mediante varios unificadores distintos.

Ejemplo

Por ejemplo, las expresiones $\varphi_1 = p(f(X), Y)$ y $\varphi_2 = p(f(a), Z)$ son unificables mediante los siguientes unificadores:

- Con $\sigma = \{X/a, Y/a, Z/a\}$, $\varphi_1\sigma = \varphi_2\sigma = p(f(a), a)$.
- Con $\sigma = \{X/a, Y/b, Z/b\}$, $\varphi_1\sigma = \varphi_2\sigma = p(f(a), b)$.
- Con $\sigma = \{X/a, Y/Z\}$, $\varphi_1\sigma = \varphi_2\sigma = p(f(a), Z)$.
- Con $\sigma = \{X/a, Z/Y\}$, $\varphi_1\sigma = \varphi_2\sigma = p(f(a), Y)$.
- ...

¿Cuál de ellos elegir? Uno que sea de **máxima generalidad** (si hay varios, cualquiera de ellos).

Definición (Unificador de máxima generalidad, u.m.g.)

Sean φ_1, φ_2 dos expresiones unificables y sea σ un unificador de $\{\varphi_1, \varphi_2\}$. Se dice que σ es un **unificador de máxima generalidad (u.m.g.)** si es *más general* que cualquier otro unificador de $\{\varphi_1, \varphi_2\}$, es decir, si para cualquier otro unificador τ resulta que existe una sustitución θ tal que $\tau = \sigma\theta$.

Ejemplo

Considere el ejemplo anterior:

- Tanto $\{X/a, Y/Z\}$ como $\{X/a, Z/Y\}$ resultan ser unificadores de máxima generalidad (intuitivamente: los más sencillos).
- Los unificadores $\{X/a, Y/a, Z/a\}$ y $\{X/a, Y/b, Z/b\}$ no puede ser u.m.g.'s puesto que no son más generales, por ejemplo, que $\{X/a, Y/Z\}$.

El Algoritmo de Unificación

- El **problema de la unificación** tiene por tanto un doble objetivo:
 - 1 Dado un par de expresiones, decidir si son o no **unificables**.
 - 2 En caso afirmativo, encontrar un **unificador de máxima generalidad**, **u.m.g.** (si hay varios, cualquiera de ellos).
- Existen distintos algoritmos capaces de resolver este problema. A continuación se presenta uno de los más comunes, basado en las ideas originales de Robinson de 1965.
- Se trata de un proceso iterativo que localiza e intenta resolver, mediante la aplicación de sustituciones adecuadas, las discordancias existentes entre las dos expresiones de entrada.
- El funcionamiento del algoritmo se ilustra a continuación con varios ejemplos, utilizando para ello una *tabla* en la que se resume la información relevante en cada uno de los pasos del algoritmo.

ALGORITMO DE UNIFICACIÓN

Entrada: Dos expresiones φ_1 y φ_2 .

Salida:

1. Un valor booleano que indica si las expresiones son unificables.
2. En caso afirmativo, un u.m.g. para φ_1, φ_2 .

Descripción:

PASO 1: $unificables := \text{cierto}$; $\sigma := \{\}$

PASO 2: MIENTRAS $unificables = \text{cierto}$ y $\varphi_1\sigma \neq \varphi_2\sigma$ HACER:

2.1. Encontrar el símbolo más a la izqda de $\varphi_1\sigma$ tal que el símbolo correspondiente de $\varphi_2\sigma$ sea distinto (en rojo en los ejemplos).

2.2. Sean s_1 y s_2 los subtérminos de $\varphi_1\sigma, \varphi_2\sigma$ que empiezan por los símbolos elegidos en el paso anterior (subrayados en los ejemplos):

(a) Si ninguno de los dos es una variable o uno es una variable que aparece en el otro (esto último se denomina **test de ocurrencia**):
 $unificables := \text{falso}$.

(b) En otro caso:

- si s_1 es una variable: $\sigma := \sigma\{s_1/s_2\}$
- si s_1 no es una variable: $\sigma := \sigma\{s_2/s_1\}$
- actualizar las expresiones $\varphi_1\sigma$ y $\varphi_2\sigma$, para lo que basta aplicar $\{s_1/s_2\}$ (o $\{s_2/s_1\}$) a las expresiones de la iteración anterior.

(Nota: $\sigma\{s_1/s_2\}$ y $\sigma\{s_2/s_1\}$ son **composiciones** de sustituciones)

PASO 3:

DEVOLVER *unificables*.

Si *unificables* = cierto, σ es un u.m.g. de φ_1, φ_2 .

Ejemplo

$\varphi_1\sigma$	$\varphi_2\sigma$	s_1	s_2	<i>unif.</i>	σ
$p(\underline{X}, X)$	$p(\underline{f(U)}, f(V))$	X	$f(U)$	cierto	$\{X/f(U)\}$
$p(f(U), f(\underline{U}))$	$p(f(U), f(\underline{V}))$	U	V	cierto	$\{X/f(V), U/V\}$
$p(f(V), f(V))$	$p(f(V), f(V))$	FIN			

El algoritmo termina en ese punto porque se llega a $\varphi_1\sigma = \varphi_2\sigma$, concluyendo que las dos expresiones de entrada son unificables con un u.m.g. dado por $\sigma = \{X/f(V), U/V\}$. Observe cómo para obtener el u.m.g. $\{X/f(V), U/V\}$ se ha realizado la *composición* de $\{X/f(U)\}$ con $\{U/V\}$. Otro posible u.m.g. sería $\{X/f(U), V/U\}$ si en la segunda línea se hubiese elegido $\{V/U\}$ en lugar de $\{U/V\}$.

Ejemplo

$\varphi_1\sigma$	$\varphi_2\sigma$	s_1	s_2	<i>unificables</i>	σ
$p(\underline{X}, f(Y))$	$p(\underline{Z}, a)$	X	Z	cierto	$\{X/Z\}$
$p(Z, \underline{f(Y)})$	$p(Z, \underline{a})$	$f(Y)$	a	falso	FIN

El Paso 2 termina en el punto **2.2. (a)**, puesto que ninguno de los dos subtérminos s_1 y s_2 es una variable, y concluye que las expresiones de entrada no son unificables.

Ejemplo

$\varphi_1\sigma$	$\varphi_2\sigma$	s_1	s_2	<i>unif.</i>	σ
$p(\underline{X}, f(X), X)$	$p(\underline{U}, W, W)$	X	U	cierto	$\{X/U\}$
$p(U, \underline{f(U)}, U)$	$p(U, \underline{W}, W)$	$f(U)$	W	cierto	$\{X/U, W/f(U)\}$
$p(U, f(U), \underline{U})$	$p(U, f(U), \underline{f(U)})$	U	$f(U)$	falso	FIN

El Paso 2 termina en el punto **2.2. (a)**, puesto que aunque el subtérmino s_1 es una variable, s_2 contiene a esa variable, por lo que falla el test de ocurrencia y se concluye que las expresiones de entrada no son unificables.

Ejercicios (Aplicación del Algoritmo de Unificación)

Para cada uno de los siguientes pares de expresiones, aplique el Algoritmo de Unificación para averiguar si las expresiones son unificables y, en caso afirmativo, facilitar un unificador de máxima generalidad (u.m.g.).

- 1 $p(W, X, f(g(Y)))$ y $p(Z, f(Z), f(U))$.
- 2 $p(a, X, f(g(Y)))$ y $p(Z, h(Z, U), f(U))$.
- 3 $q(h(X, a), X)$ y $q(Y, f(a, c))$.
- 4 $p(a, W, X, f(f(X)))$ y $p(Z, g(Y), g(Z), f(Y))$.
- 5 $q(X, g(Y), a)$ y $q(c, Z, Z)$.
- 6 $p(f(g(Y)), Y, Z)$ y $p(f(U), a, U)$.
- 7 $q(f(g(Z, Y)), Y, Z)$ y $q(f(U), b, U)$.

(los dos últimos ejercicios están resueltos en el vídeo Ejemplos Algoritmo de Unificación).

Soluciones propuestas:

1. Las expresiones $p(W, X, f(g(Y)))$ y $p(Z, f(Z), f(U))$ son unificables, como muestra la tabla con la aplicación del Algoritmo de Unificación resumida que se incluye a continuación. Recuerde que en el paso de una fila de la tabla a la siguiente la sustitución s_1/s_2 (o s_2/s_1) se aplica a las dos expresiones, φ_1 y φ_2 , *completas*, es decir, reemplazando *todas* las posibles apariciones de s_1 por s_2 , tanto en φ_1 como en φ_2 . Es el caso en el paso de la fila 1 a la fila 2 de la tabla siguiente, donde se reemplazan *todas* las apariciones de Z por W . El u.m.g. obtenido es $\{Z/W, X/f(W), U/g(Y)\}$, aunque otro u.m.g. posible sería $\{W/Z, X/f(Z), U/g(Y)\}$, dado que la primera discordancia, que involucra a dos variables, se podría resolver tanto con Z/W como con W/Z .

$\varphi_1 \sigma$	$\varphi_2 \sigma$	s_1	s_2	unif.	σ
$p(\underline{W}, X, f(g(Y)))$	$p(\underline{Z}, f(Z), f(U))$	W	Z	cierto	$\{Z/W\}$
$p(W, \underline{X}, f(g(Y)))$	$p(W, \underline{f(W)}, f(U))$	X	$f(W)$	cierto	$\{Z/W, X/f(W)\}$
$p(W, f(W), f(\underline{g(Y)}))$	$p(W, f(W), f(\underline{U}))$	$g(Y)$	U	cierto	$\{Z/W, X/f(W), U/g(Y)\}$
$p(W, f(W), f(g(Y)))$	$p(W, f(W), f(g(Y)))$	FIN			

2. Las expresiones $p(a, X, f(g(Y)))$ y $p(Z, h(Z, U), f(U))$ son unificables con u.m.g. $\{Z/a, X/h(a, g(Y)), U/g(Y)\}$. La tabla resumiendo los pasos del algoritmo en este caso se incluye a continuación. Recuerde que las sustituciones de la última columna de la tabla se obtienen *componiendo* la sustitución de la fila anterior con el nuevo par $\{s_1/s_2\}$ (o $\{s_2/s_1\}$), y que dicha composición no siempre es igual a una unión. En particular, el u.m.g. de la tercera fila es la *composición* de $\{Z/a, X/h(a, U)\}$ con $\{U/g(Y)\}$, que es igual a $\{Z/a, X/h(a, g(Y)), U/g(Y)\}$, conjunto distinto a la unión.

$\varphi_1 \sigma$	$\varphi_2 \sigma$	s_1	s_2	<i>unif.</i>	σ
$p(\underline{a}, X, f(g(Y)))$	$p(\underline{Z}, h(Z, U), f(U))$	a	Z	cierto	$\{Z/a\}$
$p(a, \underline{X}, f(g(Y)))$	$p(a, h(\underline{a}, U), f(U))$	X	$h(a, U)$	cierto	$\{Z/a, X/h(a, U)\}$
$p(a, h(a, U), f(\underline{g}(Y)))$	$p(a, h(a, U), f(\underline{U}))$	$g(Y)$	U	cierto	$\{Z/a, X/h(a, g(Y)), U/g(Y)\}$
$p(a, h(a, g(Y)), f(g(Y)))$	$p(a, h(a, g(Y)), f(g(Y)))$	FIN			

3. Las expresiones $q(h(X, a), X)$ y $q(Y, f(a, c))$ son unificables con u.m.g. $\{Y/h(f(a, c), a), X/f(a, c)\}$ (observe la *composición* en el segundo paso):

$\varphi_1 \sigma$	$\varphi_2 \sigma$	s_1	s_2	<i>unif.</i>	σ
$q(\underline{h(X, a)}, X)$	$q(\underline{Y}, f(a, c))$	$h(X, a)$	Y	cierto	$\{Y/h(X, a)\}$
$q(h(X, a), \underline{X})$	$q(h(X, a), \underline{f(a, c)})$	X	$f(a, c)$	cierto	$\{Y/h(f(a, c), a), X/f(a, c)\}$
$q(h(f(a, c), a), f(a, c))$	$q(h(f(a, c), a), f(a, c))$	FIN			

4. Como muestra la tabla siguiente, las expresiones dadas son unificables con u.m.g. $\{Z/a, W/g(f(g(a))), X/g(a), Y/f(g(a))\}$. Note en particular cómo en el paso de la fila 1 a la fila 2 se reemplazan *todas* las apariciones de Z por a y cómo en la penúltima fila se realiza la *composición* de la σ anterior con el nuevo par $\{Y/f(g(a))\}$.

$\varphi_1 \sigma$	$\varphi_2 \sigma$	s_1	s_2	<i>unif.</i>	σ
$p(\underline{a}, W, X, f(f(X)))$	$p(\underline{Z}, g(Y), g(Z), f(Y))$	a	Z	cierto	$\{Z/a\}$
$p(a, \underline{W}, X, f(f(X)))$	$p(a, \underline{g(Y)}, g(a), f(Y))$	W	$g(Y)$	cierto	$\{Z/a, W/g(Y)\}$
$p(a, g(Y), \underline{X}, f(f(X)))$	$p(a, g(Y), \underline{g(a)}, f(Y))$	X	$g(a)$	cierto	$\{Z/a, W/g(Y), X/g(a)\}$
$p(a, g(Y), g(a), f(\underline{f(g(a))}))$	$p(a, g(Y), g(a), f(\underline{Y}))$	$f(g(a))$	Y	cierto	$\{Z/a, W/g(f(g(a))), X/g(a), Y/f(g(a))\}$
$p(a, g(f(g(a))), g(a), f(f(g(a))))$	$p(a, g(f(g(a))), g(a), f(f(g(a))))$	FIN			

5. A simple vista se puede comprobar que las expresiones $q(X, g(Y), a)$ y $q(c, Z, Z)$ no son unificables, puesto que la variable Z no puede unificar a la vez con dos términos no unificables entre sí como son $g(Y)$ y a . La tabla correspondiente sería la siguiente:

$\varphi_1 \sigma$	$\varphi_2 \sigma$	s_1	s_2	<i>unif.</i>	σ
$q(\underline{X}, g(Y), a)$	$q(\underline{c}, Z, Z)$	X	c	cierto	$\{X/c\}$
$q(c, \underline{g(Y)}, a)$	$q(c, \underline{Z}, Z)$	$g(Y)$	Z	cierto	$\{X/c, Z/g(Y)\}$
$q(c, g(Y), \underline{a})$	$q(c, g(Y), \underline{g(Y)})$	a	$g(Y)$	falso	FIN

Unificación en PROLOG

PROLOG implementa el Algoritmo de Unificación anterior **omitiendo el test de ocurrencia (occur-check)** por razones de eficiencia. El lenguaje usa internamente este algoritmo (para aplicar la Regla de Resolución, ver siguiente apartado) pero también lo ofrece para uso del programador:

- El predicado predefinido **= (predicado de unificación)**, se usa en notación infija ($\text{arg1} = \text{arg2}$) y:
 - Si las dos expresiones que se le pasan resultan ser, *omitiendo el test de ocurrencia*, unificables, devuelve cierto y realiza las sustituciones de variables indicadas por el u.m.g. calculado por el Algoritmo de Unificación.
 - En caso contrario falla (devuelve *false*).
- El predicado predefinido **\= (predicado de no unificación)** también se usa en notación infija ($\text{arg1} \backslash= \text{arg2}$) y devuelve cierto (*true*) si el predicado = falla, y falla (devuelve *false*) en caso contrario.

Ejemplo (Unificación en PROLOG)

```
?- f(X, g(X, c)) = f(h(U), Z) .
```

```
      X = h(U),      Z = g(h(U), c) .
```

```
?- f(X, g(X, c)) \= f(h(U), Z) .
```

```
      false
```

```
?- p(f(X), g(Y)) = p(U, f(U)) .
```

```
      false
```

```
?- p(f(X), g(Y)) \= p(U, f(U)) .
```

```
      true
```

```
?- X = f(X) .
```

```
      X = f(X)
```

```
% el predicado = no realiza el test de ocurrencia
```

```
?- unify_with_occurs_check(X, f(X)) .
```

```
      false
```

```
% este predicado sí realiza el test de ocurrencia
```

Observación (Unificación de variables (semi)anónimas)

La variable anónima `_` y las semianónimas (`__` más todas las que empiezan por subrayado seguido de letra mayúscula) presentan las siguientes peculiaridades respecto a las variables normales:

	Normales	Semianónimas	Anónima
¿Tienen que unificar?	sí: $p(U, U) = p(a, b)$ da <i>false</i>	sí: $p(_U, _U) = p(a, b)$ da <i>false</i>	no: $p(_, _) = p(a, b)$ da <i>true</i>
¿Reportan valor?	sí: $p(U, b) = p(a, b)$ da $U = a$	no: $p(_U, b) = p(a, b)$ da <i>true</i>	no: $p(_, b) = p(a, b)$ da <i>true</i>

Ejercicios (El Algoritmo de Unificación de PROLOG)

- *Ejercicio nº 3 de la **Práctica de PROLOG nº 1** (soluciones comentadas en **Práctica de PROLOG nº 1 con soluciones**).*

LA REGLA DE RESOLUCIÓN

Definición (Regla de Resolución (1/3))

La Regla de Resolución de PROLOG recibe como entrada:

- Una cláusula objetivo (consulta) ? – $A_1, A_2, \dots, A_n, n \geq 1$.
 - Una cláusula de programa, que podrá ser un hecho B . o bien una regla $B : - B_1, \dots, B_m$. con $m \geq 1$.
- 1 Si las cláusulas de entrada tienen variables en común, se deben hacer los renombramientos de variables oportunos (por convenio, en la cláusula de programa) para evitar esta colisión de variables.
 - 2 Si el ▶ Algoritmo de Unificación determina que el *primer* subobjetivo de la cláusula objetivo, A_1 , es unificable, mediante un cierto u.m.g. σ , con la cabeza B de la cláusula de programa, entonces la Regla de Resolución se puede aplicar y se llama **cláusula resolvente** a la cláusula objetivo obtenida como sigue:

Definición (Regla de Resolución (2/3))

- Si la cláusula de programa es un **hecho** $B.$, la cláusula resolvente es

$$? - A_2\sigma, \dots, A_n\sigma$$

El subobjetivo A_1 desaparece porque se ha unificado con el hecho B mediante σ . Gráficamente:

$$\begin{array}{c}
 ? - \underline{A_1}, A_2, \dots, A_n. \\
 | \\
 \underline{B}. \quad \sigma = \{\dots, \dots\} \\
 | \\
 ? - A_2\sigma, \dots, A_n\sigma.
 \end{array}$$

Definición (Regla de Resolución (3/3))

- Si la cláusula de programa es una **regla** $B : - B_1, \dots, B_m$, la cláusula resolvente es

$$? - \underbrace{B_1\sigma, \dots, B_m\sigma}_{\text{cuerpo de la regla}}, \underbrace{A_2\sigma, \dots, A_n\sigma}_{\text{resto de subobjetivos}}$$

El subobjetivo A_1 se reemplaza por $B_1\sigma, \dots, B_m\sigma$ puesto que se ha unificado con B mediante σ . Gráficamente:

$$\begin{array}{c}
 ? - \underline{A_1}, A_2, \dots, A_n. \\
 \left| \begin{array}{l} \underline{B} : - B_1, \dots, B_m. \quad \sigma = \{ \dots, \dots \} \end{array} \right. \\
 ? - B_1\sigma, \dots, B_m\sigma, A_2\sigma, \dots, A_n\sigma.
 \end{array}$$

Observación

En las cláusulas resolventes:

- $A_i\sigma/B_i\sigma$ son el resultado de ▶ aplicar la sustitución σ a los predicados A_i/B_i .
- Se debe **respetar el orden** en el que aparecen los predicados en la cláusula resolvente (primero los $B_i\sigma$, en el orden en el que aparecen en la regla, y luego los $A_i\sigma$, en el orden en el que estaban en la cláusula objetivo original).
- Al aplicar la Regla de Resolución entre un objetivo *unitario* “ $? - A_1$.” y un hecho “ B .” unificable con A_1 , la cláusula resolvente obtenida es $?-$, que se conoce como la **cláusula vacía**.

Ejemplo (Regla de Resolución)

Dada la cláusula objetivo $?- \text{amigo}(\text{abel}, X), \text{enemigo}(X, Y).$

- La Regla de Resolución no es aplicable con el hecho $\text{enemigo}(\text{abel}, \text{cain})$ puesto que $\text{amigo}(\text{abel}, X)$ y $\text{enemigo}(\text{abel}, \text{cain})$ no son unificables (¡ni siquiera tienen el mismo nombre de predicado!).
- Tampoco con la regla $\text{amigo}(\text{abilio}, X1) :- \text{amigo}(\text{abel}, X1),$ puesto que $\text{amigo}(\text{abel}, X)$ y $\text{amigo}(\text{abilio}, X1)$ no son unificables (tienen dos constantes distintas como 1er argumento).
- Sí lo sería con $\text{amigo}(X1, Y1) :- \text{enemigo}(X1, Z1), \text{enemigo}(Z1, Y1)$ (con renombramiento de variables debido a la coincidencia de variables entre las dos cláusulas): $\text{amigo}(\text{abel}, X)$ y $\text{amigo}(X1, Y1)$ sí son unificables, con u.m.g.
 $\sigma = \{X1/\text{abel}, X/Y1\}$, y la cláusula resolvente sería $?- \text{enemigo}(\text{abel}, Z1), \text{enemigo}(Z1, Y1), \text{enemigo}(Y1, Y).$

Ejemplo (Regla de Resolución)

¿Es posible aplicar la Regla de Resolución entre el objetivo

? $\neg q(X, a), p(X), q(b, Y)$ y la regla $q(f(X), Y) :- r(X), r(Y)$?

- 1 Como las dos cláusulas dadas tienen variables en común, se renombran las variables de la regla, que pasa a ser $q(f(X1), Y1) :- r(X1), r(Y1)$.
- 2 la Regla de Resolución sí es aplicable entre las cláusulas dadas puesto que el Algoritmo de Unificación muestra que el primer predicado del objetivo, $q(X, a)$, es unificable con la cabeza de la regla, $q(f(X1), Y1)$, mediante u.m.g. $\sigma = \{X/f(X1), Y1/a\}$. La cláusula resolvente correspondiente es entonces:

$$\begin{aligned} & ? - r(X1)\sigma, r(Y1)\sigma, p(X)\sigma, q(b, Y)\sigma \\ = & ? - r(X1), r(a), p(f(X1)), q(b, Y). \end{aligned}$$

Ejercicios (Aplicación de la Regla de Resolución)

Para cada uno de los siguientes pares de cláusulas, razone si es posible aplicar la Regla de Resolución, y, en caso afirmativo, calcule la cláusula resolvente asociada:

- ❶ *Objetivo ? – $r(Z, a), q(a, f(a))$. y regla $q(X, X) : \neg p(X), r(X, a)$.*
- ❷ *Objetivo ? – $q(a, f(a)), r(Z, a)$. y regla $q(X, X) : \neg p(X), r(X, a)$.*
- ❸ *Objetivo ? – $q(X, a), p(X), q(b, Y)$. y hecho $q(X, X)$.*
- ❹ *Objetivo ? – $r(f(X), X), q(X, Z), r(f(Z), f(f(a)))$. y regla $r(X, f(Z)) : \neg p(X, Z), q(Z, X)$.*
- ❺ *Objetivo ? – $p(X, b)$. y hecho $p(h(a, X), X)$.*

Soluciones propuestas:

1. No es posible aplicar la Regla de Resolución, ya que los predicados $r(Z, a)$ y $q(X, X)$ no son unificables (ni siquiera tienen mismo nombre de predicado).
2. No es posible aplicar la Regla de Resolución, ya que los predicados $q(a, f(a))$ y $q(X, X)$ no son unificables (la variable x no puede unificar al mismo tiempo con dos términos, a y $f(a)$, que no son unificables entre sí).
3. Sí es posible aplicar la Regla de Resolución: renombrando el hecho a $q(X1, X1)$., mediante u.m.g. $\{X1/a, X/a\}$ se obtiene la cláusula resolvente
 $?- p(a), q(b, Y)$.

4. Sí es posible aplicar la Regla de Resolución, pero es necesario renombrar la regla a

$$r(X1, f(Z1)) :- p(X1, Z1), q(Z1, X1) .$$

porque en caso contrario el test de ocurrencia del Algoritmo de Unificación impediría la unificación. El u.m.g. es $\{X1/f(f(Z1)), X/f(Z1)\}$ (ojo a la composición entre $\{X1/f(X)\}$ y $\{X/f(Z1)\}$) y se obtiene la cláusula resolvente

$$?- p(f(f(Z1)), Z1), q(Z1, f(f(Z1))), q(f(Z1), Z), \\ r(f(Z), f(f(a))) .$$

5. Sí es posible aplicar la Regla de Resolución: renombrando el hecho a $p(h(a, X1), X1)$ (si no, el test de ocurrencia impediría la unificación) y mediante u.m.g. $\{X/h(a, b), X1/b\}$ (ojo a la composición) se obtiene como cláusula resolvente la cláusula vacía

$$?-$$

EL ÁRBOL DE RESOLUCIÓN

Dados:

- Un *programa* (conjunto de fórmulas -cláusulas de Horn positivas- que pueden ser hechos o reglas),
- Y una *consulta* (fórmula denominada cláusula de Horn negativa),

el **Árbol de Resolución** asociado a ese programa y esa consulta

- Es una herramienta para *explorar y organizar todas las posibles formas en las que es posible resolver la consulta*, es decir, los posibles caminos (si los hay) para demostrar que la consulta es una consecuencia lógica del programa y, en su caso, *computar* las soluciones asociadas.
- Se construye aplicando reiteradamente la *Regla de Resolución* como se describe a continuación.

Construcción de un Árbol de Resolución (1/2)

El Árbol de Resolución correspondiente a un programa y una consulta es un árbol etiquetado y con raíz:

- **Raíz** del árbol: la consulta.
- **Nodos**: son cláusulas objetivo de la forma “ $?- \text{ } \phi_1, \dots, \phi_k.$ ”, con $k \geq 0$, que se expanden de la siguiente forma:
 - Si $k = 0$ (cláusula vacía), el nodo no se puede expandir, es una hoja del árbol denominada **nodo éxito**.
 - Si $k \neq 0$ pero no existe ninguna cláusula en el programa con la que se pueda aplicar la **Regla de Resolución**, el nodo tampoco se puede expandir, es una hoja del árbol denominada **nodo fallo**.
 - En cualquier otro caso, el nodo tendrá un nodo hijo por cada cláusula del programa con la que se pueda aplicar la **Regla de Resolución**, de forma que los hijos contienen las *cláusulas resolventes* obtenidas al aplicar la Regla de Resolución.

Construcción de un Árbol de Resolución (2/2)

- Los **hijos** de los nodos que los tengan se colocan de izquierda a derecha eligiendo las cláusulas (hechos o reglas) del programa *de acuerdo con el orden en el que aparecen en él* (el hijo de más a la izquierda será la cláusula resolvente obtenida con la primera cláusula del programa con la que se pueda aplicar la Regla de Resolución, etc).
- Las **aristas** se etiquetan con la siguiente información:
 - El u.m.g. σ_j utilizado al aplicar la Regla de Resolución. Convenio: los u.m.g.'s se numeran en función del nivel del árbol (σ_1 en el primer nivel, etc).
 - Opcionalmente, la cláusula C_i (C_i^{ren} si se ha renombrado) del programa con la que se aplica la Regla de Resolución. Convenios: (1) la primera cláusula del programa es C_1 , la segunda C_2 , etc. (2) en caso necesario, las variables se renombran añadiéndoles un subíndice que indica el nivel del árbol (X_1 en el primer nivel, etc).

Interpretación de las hojas de un Árbol de Resolución

• Nodos éxito

- Prueban que la consulta se ha podido realizar con éxito (se ha probado que es una consecuencia lógica del programa).
- Cada nodo éxito proporciona una solución a la consulta:
 - Si la consulta no tiene variables: **true**.
 - En otro caso: la solución está compuesta por el valor $X\sigma_1 \dots \sigma_r$ para cada variable X (salvo anónimas o semianónimas) que aparezca en la consulta, donde $\sigma_1, \dots, \sigma_r$ son los u.m.g.'s asociados con cada una de las aristas de la rama, en orden, empezando desde la raíz del árbol. La forma más fácil de computar $X\sigma_1 \dots \sigma_r$ es ▶ aplicar σ_1 a X , a continuación ▶ aplicar σ_2 a la expresión resultante de la aplicación anterior, etc.

• Nodos fallo

- Son fracasos a la hora de intentar probar la consulta.
- Su aparición no significa que la consulta no se pueda realizar con éxito, simplemente muestra que la rama elegida para intentarlo no lo permite.

Ejemplo (Árboles de Resolución programa “Abuelas/os” 1/2)

`progenitor(pepa, pepito).` % C1
`progenitor(pepito, pepita).` % C2
`progenitor(pepito, pepon).` % C3
`abuelo(X,Z) :- progenitor(X,Y), progenitor(Y,Z).` % C4

? - abuelo(pepa, N).

C4 $\sigma_1 = \{X/pepa, N/Z\}$

? - progenitor(pepa, Y), progenitor(Y, Z).

C1 $\sigma_2 = \{Y/pepito\}$

? - progenitor(pepito, Z).

$\sigma_3 = \{Z/pepita\}$

C2

C3

$\sigma_3 = \{Z/pepon\}$

? -

? -

$\underline{N\sigma_1\sigma_2\sigma_3} = \underline{Z\sigma_2\sigma_3} = \underline{Z\sigma_3} = \text{pepita}$

$\underline{N\sigma_1\sigma_2\sigma_3} = \underline{Z\sigma_2\sigma_3} = \underline{Z\sigma_3} = \text{pepon}$

Ejemplo (Árboles de Resolución programa “Abuelas/os” 2/2)

? – abuelo(pepa, pepon).

| C4 $\sigma_1 = \{X/pepa, Z/pepon\}$

? – progenitor(pepa, Y), progenitor(Y, pepon).

| C1 $\sigma_2 = \{Y/pepito\}$

? – progenitor(pepito, pepon).

| C3 $\sigma_3 = \{\}$

? –

true

? – abuelo(pepa, pepito).

| C4 $\sigma_1 = \{X/pepa, Z/pepito\}$

? – progenitor(pepa, Y), progenitor(Y, pepito).


| C1 $\sigma_2 = \{Y/pepito\}$

? – progenitor(pepito, pepito).



fallo

Al construir Árboles de Resolución en los que aparezcan variables anónimas o semianónimas se deben tener en cuenta las [particularidades](#) de este tipo de variables en la unificación. Compare los tres árboles a continuación.

P = variable normal




Ejemplo (Árboles con variables anónimas/semianónimas (2/2))



Ejemplo (Árbol de Resolución programa “suma”)

$\text{suma}(X, 0, X) . \quad \% \text{ C1}$

$\text{suma}(X, s(Y), s(Z)) :- \text{suma}(X, Y, Z) . \quad \% \text{ C2}$



Observación (Árboles con ramas infinitas)


Los Árboles de Resolución también pueden tener **ramas infinitas**, que pueden ser de dos clases:

- Ramas con un número infinito de soluciones que van apareciendo sucesivamente (ver ejemplo a continuación con el programa de la suma), o bien
- Ramas sin soluciones que desarrollan hijos indefinidamente, hasta que acaban desbordando la memoria (ver ejemplo más adelante con la tercera versión del programa “ancestro”,
▶ `ancestro3`, rama de más a la derecha).

Ejemplo (Árbol de Resolución con infinitas soluciones)

suma (X, 0, X) . % C1

suma (X, s (Y) , s (Z)) :- suma (X, Y, Z) . % C2



Ejercicios (Construcción de Árboles de Resolución)

Construya los Árboles de Resolución asociados con los programas y consultas dados a continuación:

- 1 ▶ *Programa Abuelos* y consulta (escribala en LPO y en PROLOG) adecuada para averiguar quiénes son abuelas/os de Pepón.
- 2 ▶ *Programa Abuelos* y consultas (escribalas en LPO y en PROLOG) adecuadas para averiguar si Pepito y Pepón son hermanos (tienen algún progenitor en común) y, en caso afirmativo, saber quiénes son esos progenitores comunes. ¿Y Pepita y Pepón?
- 3 *Programa Amistades/Enemistades* y consulta (escribala en LPO y en PROLOG) adecuada para averiguar si Abel es amigo de alguien que, a su vez, sea enemigo de alguien, y, en caso afirmativo, saber quiénes serían.
- 4 ▶ *Programa Suma* y consulta ?- `suma(X, Y, s(s(0)))`. (escribala en LPO y describa su objetivo en lenguaje natural).

Soluciones propuestas:

1.

? - abuelo(A, pepon).

C4 $\sigma_1 = \{A/X, Z/pepon\}$

? - progenitor(X, Y), progenitor(Y, pepon).

$\sigma_2 = \{X/pepa, Y/pepito\}$ C1

C3 $\sigma_2 = \{X/pepito, Y/pepon\}$

C2 $\sigma_2 = \{X/pepito, Y/pepita\}$

? - progenitor(pepito, pepon).

$\sigma_3 = \{ \}$ C3

? -

? - progenitor(pepita, pepon).

fallo

? - progenitor(pepon, pepon).

fallo

$\underline{A\sigma_1\sigma_2\sigma_3 = X\sigma_2\sigma_3 = pepa\sigma_3 = pepa}$

2a)

? – progenitor(X, pepito), progenitor(X, pepon).

C1 $\sigma_1 = \{X/pepa\}$

? – progenitor(pepa, pepon).

fallo

2b)

? – progenitor(X, pepita), progenitor(X, pepon).

C2 $\sigma_1 = \{X/pepito\}$


? – progenitor(pepito, pepon).

C3 $\sigma_2 = \{a\}$


? –

$X\sigma_1\sigma_2 = pepito\sigma_2 = pepito$

3.



4. LPO $\exists X \exists Y \text{suma}(X, Y, s(s(0)))$. ¿Qué formas hay de descomponer el natural 2 en dos sumandos?




Mecanismo de cómputo de PROLOG


Cuando recibe una consulta relativa a un programa, PROLOG **construye el Árbol de Resolución** correspondiente de la siguiente forma:

- El árbol se construye en **profundidad** por la izquierda (ver dibujo en la página siguiente).
- El árbol se construye utilizando **retroceso (backtracking)** automático al llegar a un nodo fallo: cuando se encuentra con uno de estos nodos, PROLOG *no da cuenta del fallo*, simplemente retrocede para continuar la búsqueda (en profundidad) de posibles éxitos.
- El árbol **no se construye necesariamente entero**: en cada nodo *éxito* el sistema pregunta al usuario si quiere seguir o terminar la ejecución de la consulta.

Ejemplo de orden (en profundidad) con el que PROLOG construye los Árboles de Resolución:



Actitud de PROLOG al construir (en profundidad) un Árbol de Resolución



Ejemplo (Respuestas ofrecidas por PROLOG)

¿Qué respuestas ofrecería PROLOG, y en qué orden, al construir los Árboles de Resolución, completos, de los últimos ejemplos?

- En los ejemplos ▶ Árboles de Resolución programa Abuelas/os :
 - En el primer árbol: primera solución: $N = \text{pepita}$, segunda solución: $N = \text{pepon}$ y a continuación `false` (este último indicando que no hay más soluciones).
 - Las respuestas para los otros dos árboles serían `true` y `false`.
- En el ejemplo ▶ Árbol de Resolución programa Suma :
 - Primera solución: $Z = s(0)$
 - `false` (indicando que no hay más soluciones -el nodo fallo no se reporta-)
- En el ejemplo ▶ Árbol de Resolución con infinitas soluciones :
 - Primera solución: $U = X, V = 0, W = X$
 - Segunda solución: $U = X2, V = s(0), W = s(X2)$
 - Tercera solución: $U = X3, V = s(s(0)), W = s(s(X3))$
 - Cuarta solución: ... (seguiría dando todas las posibles soluciones)

Ejercicios (Respuestas ofrecidas por PROLOG)

Para cada uno de los Árboles construidos en los ejercicios

► *Construcción de Árboles de Resolución*, indique qué respuestas ofrecería PROLOG, y en qué orden lo haría.

Soluciones propuestas:

1. "Abuelas/os", ?- abuelo(A, pepon) .
 - Primera solución: A = pepa
 - False (indicando que no hay más soluciones, los nodos fallo no se reportan).
- 2a. "Abuelas/os", ?- progenitor(X, pepito), progenitor(X,pepon) .
 - False (indicando que no hay soluciones)
- 2b. "Abuelas/os", ?- progenitor(X, pepita), progenitor(X,pepon) .
 - Primera solución: X = pepito
 - False (indicando que no hay más soluciones)
3. "Amistades/Enemistades", ?- amigo(abel,X), enemigo(X,Y) .
 - Primera solución: X = blas, Y = abilio
 - False (indicando que no hay más soluciones, los nodos fallo no se reportan)
4. "Suma", ?- suma(X, Y, s(s(0))) .
 - Primera solución: X = s(s(0)), Y = 0
 - Segunda solución: X = s(0), Y = s(0)
 - Tercera solución: X = 0, Y = s(s(0))
 - False (indicando que no hay más soluciones)

Observación (Influencia del orden en programas y reglas)

Tanto:

- 1 El orden de las cláusulas en el programa, como
- 2 El orden de los predicados en el cuerpo de una regla.

pueden influir en:

- 1 Qué soluciones se encuentran y en qué orden aparecen.
- 2 La terminación o no de las consultas que se realizan.

En particular, **la búsqueda de soluciones no es completa**: no asegura encontrar una solución a la consulta aunque la haya (esto ocurrirá siempre que el Árbol de Resolución tenga alguna rama infinita más a la izquierda que la primera rama éxito).

El predicado `ancestro`, cuya implementación se discute a continuación, se usa a menudo para ilustrar lo anterior.

Ejemplo (El predicado `ancestro`)

Dado `progenitor(X, Y)`, cierto si X es progenitor/a de Y , se define el predicado `ancestro(X, Y)`, cierto si X es un ancestro de Y , es decir, si X es progenitor/abuela-o/bisabuela-o/etc de Y .

- Caso base. Los progenitores son ancestros de sus hijas/os:

$\forall X \forall Y (\text{progenitor}(X, Y) \rightarrow \text{ancestro}(X, Y))$. En Prolog:

```
ancestro(X, Y) :-
    progenitor(X, Y).
```

- Caso recursivo. Los progenitores son ancestros de todos aquellos de los que sus hijas/os lo son.

$\forall X \forall Y \forall Z ((\text{progenitor}(X, Z) \wedge \text{ancestro}(Z, Y)) \rightarrow \text{ancestro}(X, Y))$.

En Prolog:

```
ancestro(X, Y) :-
    progenitor(X, Z),
    ancestro(Z, Y).
```

Ejemplo (Influencia del orden en programas y reglas)

```
progenitor(pepa, pepito).
```

```
progenitor(pepito, pepon).
```

```
% CUATRO POSIBLES ORDENACIONES PARA "ancestro"
```

```
(todas equivalentes en Lógica, pero no en Prolog)
```

```
ancestro1(X,Y) :- progenitor(X,Y).
```

```
ancestro1(X,Y) :- progenitor(X,Z), ancestro1(Z,Y).
```

```
ancestro2(X,Y) :- progenitor(X,Z), ancestro2(Z,Y).
```

```
ancestro2(X,Y) :- progenitor(X,Y).
```


```
ancestro3(X,Y) :- progenitor(X,Y).
```

```
ancestro3(X,Y) :- ancestro3(Z,Y), progenitor(X,Z).
```


```
ancestro4(X,Y) :- ancestro4(Z,Y), progenitor(X,Z).
```

```
ancestro4(X,Y) :- progenitor(X,Y).
```

a1 = ancestro1, pg = progenitor



a3 = ancestro3, pg = progenitor



Los árboles de ancestro1 y ancestro3 son muy distintos entre sí. Los de ancestro2 y ancestro4 no son otra cosa que las *imágenes especulares* de los árboles de ancestro1 y ancestro3, respectivamente (hágalos como ejercicio).

Respuestas ofrecidas: PROLOG construye los Árboles de Resolución correspondientes en profundidad, ofreciendo las soluciones de las ramas éxito según las va encontrando y sin reportar las ramas fallo (donde hace backtracking de forma automática):

```
?- ancestro1(pepa, D).
```

```
D = pepito ; D = pepon
```

```
?- ancestro2(pepa, D).
```

```
D = pepon ; D = pepito. % Soluciones intercambiadas
```

```
?- ancestro3(pepa, D).
```

```
D = pepito ; D = pepon ; % Encuentra las soluciones
```

```
ERROR: Out of local stack % rama infinita
```

```
?- ancestro4(pepa, D). % ;NO encuentra soluciones!
```

```
ERROR: Out of local stack % rama infinita
```

- No existe ninguna regla general que establezca el orden óptimo de las cláusulas en el programa ni el orden óptimo de los predicados en el cuerpo de las reglas (depende de cada caso).
- En general, sí es recomendable empezar por “lo más sencillo”:
 - 1 Colocar primero los casos base (cláusulas que expresan las condiciones de parada de la recursividad).
 - 2 Evitar las reglas con recursión a la izquierda (reglas tales que el predicado del primer subobjetivo de su cuerpo coincide con el predicado de su cabeza), primando siempre que sea posible la recursión de cola o recursión final (recursión en último lugar).

Ejemplo

De las cuatro versiones del predicado `ancestro`, la única que cumple las dos recomendaciones anteriores es la primera, `ancestral`.

Ejercicios (El orden importa + uso del depurador)

Ejercicio nº 4 de la *Práctica de PROLOG nº 1*.

Ejercicios (Mecanismo de cómputo de PROLOG)

1 Considere el programa

```
p(X, X) . % C1
p(X, Z) :- q(X, Y) , p(Y, Z) . % C2
q(a, b) . % C3
```

- 1 *Expresa en lenguaje natural y en LPO el significado de cada una de las cláusulas del programa.*
- 2 *Construya el Árbol de Resolución correspondiente a la consulta $?- p(X, b)$. ¿Qué se pretende averiguar? ¿Qué respuestas ofrecería PROLOG, y en qué orden?*
- 3 *¿Qué consecuencias tendría el intercambio de las dos primeras cláusulas del programa?*
- 4 *¿Qué consecuencias tendría el intercambio de los dos predicados del cuerpo de la regla?*
- 5 *¿Qué significado tiene la consulta $?- p(X, b) , q(X, b)$? Construya el Árbol de Resolución necesario e indique qué respondería PROLOG ante esta consulta.*

Soluciones propuestas:

1. El programa define dos relaciones binarias, $p/2$ y $q/2$, de la siguiente forma:


- C1: todo objeto está relacionado consigo mismo mediante la relación $p/2$ (es decir, la relación p es reflexiva). LPO: $\forall X p(X, X)$.
- C2: para cualesquiera objetos X, Y, Z , si X está relacionado mediante q con Y y ese Y está relacionado con Z mediante p , entonces X y Z están relacionados mediante p . LPO: $\forall X \forall Y \forall Z ((q(X, Y) \wedge p(Y, Z)) \rightarrow p(X, Z))$.
- El objeto a está relacionado con el objeto b mediante q . LPO: $q(a, b)$.

2. La consulta $?- p(X, b)$ pretende averiguar si existe algún objeto relacionado con b mediante p , y, en caso afirmativo, qué objeto u objetos cumplen lo anterior. En LPO: $\exists X p(X, b)$. Ver Árbol y soluciones en página siguiente.

3. Si se intercambian las dos primeras cláusulas, el Árbol de Resolución sería la imagen especular (imagen generada al reflejarse sobre un espejo) del árbol del apartado anterior, por lo que las soluciones aparecerían intercambiadas: la primera solución sería $X=a$ y la segunda $X=b$.

4. Si se intercambian los predicados del cuerpo de la regla, se pasa de una recursión de cola o final (a la derecha) a una recursión a la izquierda, y la estructura del Árbol cambia. Este Árbol está dibujado más adelante y también puede encontrarse una descripción detallada de su construcción [aquí](#).


2.



PROLOG construye el Árbol de Resolución anterior en profundidad, haciendo backtracking automático en las ramas fallo (sin reportar estos fallos), por lo que las soluciones que ofrece ante la consulta dada son:

- Primera solución: $X = b$
- Segunda solución: $X = a$
- False (indicando que no hay más soluciones)


4.



A la vista del árbol anterior, la respuesta de PROLOG será la siguiente:

- Primera solución: $X = b$
- Segunda solución: $X = a$
- ERROR: Out of local stack (por computación infinita)

5. ¿Qué objeto(s) existen(n) que estén relacionado con b tanto mediante p como mediante q ?



PROLOG construye el Árbol de Resolución anterior en profundidad, haciendo backtracking automático en las ramas fallo (sin reportar estos fallos):

- Primera solución: $X = a$
- False (indicando que no hay más soluciones)

BIBLIOGRAFÍA

- L. Sterling and E. Shapiro. *The Art of Prolog*. The MIT Press, Cambridge, Mass., second edition, 1994.
- W.F. Clocksin and C.S. Mellish. *Programming in Prolog*. Springer-Verlag, Berlin, fifth edition, 2003.
- I. Bratko. *Prolog Programming for Artificial Intelligence*. Addison-Wesley, Reading, Massachusetts, third edition, 2001.
- J. Lloyd. *Foundations of Logic Programming*, (Second Edition). Springer-Verlag, 1987.
- R. O'Keefe. *The Craft of Prolog*. The MIT Press, Cambridge, MA, 1990.
- U. Nilsson and J. Maluszynski. **Logic, Programming and Prolog**. John Wiley & Sons Ltd, 1996.
- **SWI-Prolog**, entorno de programación en Prolog de dominio público.
- **comp.lang.prolog. Faq**

© 2022 Ana Pradera Gómez

Algunos derechos reservados

Este documento se distribuye bajo la licencia

“Atribución-CompartirIgual 4.0 Internacional” de Creative Commons,
disponible en

<https://creativecommons.org/licenses/by-sa/4.0/deed.es>